*---------------------------------------------------Reproduction files for----------------------------------------------------------------*
*                                   Laustsen 2017: Choosing the Right Candidate:
* Observational and Experimental Evidence that Conservatives and Liberals Prefer Powerful and Warm Candidate Personalities, Respectively
*                                                    Political Behavior

***********************************Study 1: AMERICAN NATIONAL ELECTION STUDIES*************************************************************

***** Excludes data that does not contain simultaneous trait ratings of power and warmth of presidential candidates
drop if VCF0004 < 1984
drop if VCF0004 == 1986
drop if VCF0004 == 1990
drop if VCF0004 == 1994
drop if VCF0004 == 1998
drop if VCF0004 == 2002
drop if VCF0004 == 2012


**************************************************RECODES DIFFERENT VARIABLES**************************************************************
*****Respondents' ideological self-placement on 7-point scale (1 = Extremely liberal; 7 = Extremely conservative)
recode VCF0803 (0 9 =.), generate(libcons7)
generate ideology = (libcons7-1)/6


*****Trait perceptions of political candidates - coded to 0-1 scales with "0" and "1" constituting min. and max vaklues, respectively. 
*Republican candidates are coded "1" and Democrats are coded "2"
// Categories for trait variables:
*1. Extremely well
*2. Quite well
*3. Not too well
*4. Not well at all

*Missing codes:
*8: DK
*9: NA
*0:

*** Trait: Strong leadership
recode VCF0368 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(lead1)
recode VCF0356 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(lead2)
**0-1 variables
generate Lead1 = (lead1-1)/3
generate Lead2 = (lead2-1)/3


*** Trait: 'Really cares about people like you'
recode VCF0369 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(care1)
recode VCF0357 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(care2)
**0-1 variables
generate Care1 = (care1-1)/3
generate Care2 = (care2-1)/3


*** Trait: Compassion
recode VCF0363 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(comp1)
recode VCF0351 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(comp2)
**0-1 variables
generate Comp1 = (comp1-1)/3
generate Comp2 = (comp2-1)/3


*** Trait: Moral
recode VCF0367 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(moral1)
recode VCF0355 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(moral2)
**0-1 variables
generate Moral1 = (moral1-1)/3
generate Moral2 = (moral2-1)/3


*** Trait: Intelligent
recode VCF0362 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(intel1)
recode VCF0350 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(intel2)
**0-1 variables
generate Intel1 = (intel1-1)/3
generate Intel2 = (intel2-1)/3

*** Trait: Decent
recode VCF0364 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(decent1)
recode VCF0352 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(decent2)
**0-1 variables
generate Decent1 = (decent1-1)/3
generate Decent2 = (decent2-1)/3

*** Trait: Inspiring
recode VCF0353 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(inspir1)
recode VCF0351 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(inspir2)
**0-1 variables
generate Inspir1 = (inspir1-1)/3
generate Inspir2 = (inspir2-1)/3

*** Trait: Knowledgeable
recode VCF0366 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(know1)
recode VCF0354 (1=4) (2=3) (3=2) (4=1) (0 = .) (8 =.) (9 =.), generate(know2)
**0-1 variables
generate Know1 = (know1-1)/3
generate Know2 = (know2-1)/3



*** Creates common WARMTH variable based on 'Really cares about people like you' and Compassion
egen Warmth1 = rowmean(Care1 Comp1)
egen Warmth2 = rowmean(Care2 Comp2)


*** Creates common COMPETENCE variable based on intelligent and knowledgeable
egen Competence1 = rowmean(Intel1 Know1)
egen Competence2 = rowmean(Intel2 Know2)



******* DEPENDENT VARIABLES

****** Vote choice
*VALID_CODES:
*------------
*1. Democrat
*2. Republican
*3. Major third party candidate (Wallace 1968/Anderson 1980/Perot 1992,1996)

** Excludes respondents who did not vote for either the Reublican or the Democratic candiadte (see note 7 in main text)
recode VCF0704 (0 3 =.), generate(vote_choice)

** Voting for Republican candidate:
recode vote_choice (1 = 0) (2=1), generate(vote1)

** Voting for Democratic candidate:
recode vote_choice (1=1) (2 = 0), generate(vote2)



****** Feeling themometers

** Feeling thermometer rating of Republican candidate:
recode VCF0426 (98 =.) (99=.), generate(feel1)

** Feeling thermometer rating of Democratic candidate:
recode VCF0424 (98 =.) (99=.), generate(feel2)

* Creates 0-1 variables ("0" and "1" indicate coldest and warmest feelings, respectively)
generate Feel1 = feel1/100
generate Feel2 = feel2/100




****** BACKGROUND VARIABLES AND CONTROLS
**** Respondents' party identification (7-point scale recoded to 0-1 scale format;
* "0" and "1" represents "Strong Democrats" and "Strong Republicans", respectively.
recode VCF0301  (1 = 1) (2 = 2) (3 = 3) (4 = 4) (5 = 5) (6 =6) (7=7) (else =.), generate(party_7)
generate party7 = (party_7-1)/6

**** Gender: VCF0104
recode VCF0104 (1=1 "male") (2=0 "female") (else =.), generate(gender) label(genderLB)

**** Age: VCF0101
recode VCF0101 (00 97 98 99 = .), generate(age)

**** Income: summary household income - VCF0114 
//coded in percentiles: 0-16; 17-33; 34-67 68-95; 96-100.
recode VCF0114 (1=0) (2=1) (3=2) (4=3) (5=4) (0=.), generate(income)

**** Education: VCF0110
//coded in four categories
recode VCF0110 (1=0) (2=1) (3=2) (4=3) (0=.), generate(education)

**** Race: VCF0106
//coded 1 = white, 0 = nonwhite
recode VCF0106 (1=1) (2 3 = 0) (0 9 =.), generate(race)

**** Church attendance: VCF0130
// coded into five continuous categories
recode VCF0130 (5 7 = 0) (4 = 1) (3 = 2) (2 = 3) (1 = 4) (0 8 9 =.), generate(church)




**--------------------------------------------------Restructuring the dataset-------------------------------------------------------**

***** Creates unique identifier for each respondent 
generate identity = _n


reshape long Lead Care Comp Moral Intel Inspir Decent Know Warmth Competence Feel vote, i(identity) j(cand_party)

***Creates dummy variable for candidate party: 0 = Rep; 1 = Dem
recode cand_party (1=0) (2=1)


***Sets clustering variable
xtset identity

**-------------------------------------------------------------------------------------------------------------------------------------**



*------------------------------------DESCRIPTIVE STATISTICS FOR KEY VARIABLES 1984-2008-------------------------------------------------*

summ Lead Warmth Competence Moral

summ Feel

summ ideology



******************************MAIN MODEL: Interaction between power and warmth, respectively, with ideology***************************

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

*A) Feelings: Trait X Ideology interaction (Table 1, Model A) 
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe

*Graphical illustration: Figure 1, panel A
margins, dydx(Lead Warmth) at(ideology=(0 (0.05) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of cand. traits on feeling therm. ratings) title("") scheme(s2color)

*B) Vote choice: Trait X Ideology interaction (Table 1, Model B)
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, or fe

*Graphical illustration: Figure 1, panel B
margins, dydx(Lead Warmth) at(ideology=(0 (0.05) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of cand. traits on vote choice) title("") scheme(s2color)


******************************************************ROBUSTNESS TESTS***************************************************************

******ROBUSTNESS TEST 1: Controlling for threeway interaction between candiadte party, respondent party affiliation and trait:
// Reported in Online Supplementary Materials S.2

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

*A) Feelings: Trait X Ideology interaction controlling for threeway interaction (Table S.2.1., Model A)
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.cand_party##c.party7##c.Warmth i.cand_party##c.party7##c.Lead i.cand_party##ib(2008).VCF0004, fe

*Graphical illustration: Figure S.2.1
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of cand. traits on feeling therm. ratings) title("") scheme(s2color)

*B) Vote choice: Trait X Ideology interaction controlling for threeway interaction (Table S.2.1., Model B)
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.cand_party##c.party7##c.Warmth i.cand_party##c.party7##c.Lead i.cand_party##ib(2008).VCF0004, fe
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.cand_party##c.party7##c.Warmth i.cand_party##c.party7##c.Lead i.cand_party##ib(2008).VCF0004, or fe

*Graphical illustration: Figure S.2.2
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of cand. traits on vote choice) title("") scheme(s2color)



******ROBUSTNESS TEST 2: Controlling for policy inference based on trait impression of candidates
// Reported in ONLINE SUPPLEMENTARY MATERIALS S.3

*****Strong Leadership evaluations and 'hawkish foreign policy attitudes':

** Creates measures of 'Hawkish foreign policy attitudes': 0-1 scales with "1" indicating most 'compassionate' position.
*Defense spending (included 1984-2008) (VCF0843)
recode VCF0843 (0 9 = .), generate(def_spending)
generate defspending = (def_spending-1)/6

*US' willingness to use their military force (included 1992-1996): (VCF0844)
recode VCF0844 (1=5) (2=4) (3=3) (4=2) (5=1) (8 9 = .), generate(US_force)
generate USforce = (US_force-1)/4

*US cooperation with the USSR (included 1984-1988): (VCF0841)
recode VCF0841 (0 9 = .), generate(USSR_cooperate)
generate USSRcooperate = (USSR_cooperate-1)/6


***Feeling thermometer ratings - reported in Figure S.3.1.a and Table S.3.1.a

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

*Controling for rival interaction between "Defense spending" (included 1984-2008) (VCF0843) and ratings of strong leadership
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Lead##c.defspending c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "US' willingness to use their military force" (included 1992-1996) (VCF0844) and ratings of strong leadership
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Lead##c.USforce c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "US cooperation with the USSR" (included 1984-1988) (VCF0841) and ratings of strong leadership
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Lead##c.USSRcooperate c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)


**Vote Choice - reported in Figure S.3.1.b and Table S.3.1.b

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

*Controling for rival interaction between "Defense spending" (included 1984-2008) (VCF0843) and ratings of strong leadership
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Lead##c.defspending c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "US' willingness to use their military force" (included 1992-1996) (VCF0844) and ratings of strong leadership
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Lead##c.USforce c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "US cooperation with the USSR" (included 1984-1988) (VCF0841) and ratings of strong leadership
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Lead##c.USSRcooperate c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)


*****Warmth evaluations and 'compassionate and protective social policies':

** Creates measures of 'Compassionate and protective social policies': 0-1 scales with "1" indicating most 'compassionate' position.
*Government help to blacks (VCF0830)
recode VCF0830 (1=7) (2=6) (3=5) (4=4) (5=3) (6=2) (7=1) (0 9 = .), generate(gov_help)
generate govhelp = (gov_help-1)/6

*Equal roles for women and men (VCF0834)
recode VCF0834 (1=7) (2=6) (3=5) (4=4) (5=3) (6=2) (7=1) (0 9 = .), generate(equal_role)
generate equalrole = (equal_role-1)/6

*Government spending on services (VCF0839)
recode VCF0839 (0 9 = .), generate(gov_service)
generate govservice = (gov_service-1)/6


***Feeling thermometer ratings - reported in Figure S.3.2.a Table S.3.2.a

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

**Feeling thermometer Ratings
*Controling for rival interaction between "Government help to blacks" (VCF0830) and ratings of warmth
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Warmth##c.govhelp c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "Equal roles for women and men" (VCF0834) and ratings of warmth
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Warmth##c.equalrole c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "Government spending on services" (VCF0839) and ratings of warmth
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Warmth##c.govservice c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)


**Vote Choice - reported in Figure S.3.2.b Table S.3.2.b

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

*Controling for rival interaction between "Government help to blacks" (VCF0830) and ratings of warmth
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Warmth##c.govhelp c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "Equal roles for women and men" (VCF0834) and ratings of warmth
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Warmth##c.equalrole c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)

*Controling for rival interaction between "Government spending on services" (VCF0839) and ratings of warmth
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Warmth##c.govservice c.Competence c.Moral i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
*Graphical illustration:
margins, dydx(Lead Warmth) at(ideology=(0 (0.10) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of personality traits on feelings) title("") scheme(s2color)



******ROBUSTNESS TEST 3: Controlling for simultaneous interactions between evaluations of competence and moral, respectively, with ideology
// Reported in ONLINE SUPPLEMENTARY MATERIALS S.4

*----------- PLEASE BE AWARE THAT IT MIGHT TAKE SEVERAL MINUTES FOR STATA TO CREATE THE DIFFERENT MARGINAL EFFECT PLOTS ---------------*

*** Predicting feeling thermometer ratings: Reported in Table S.4.1, Model A) 
xtreg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence##c.ideology c.Moral##c.ideology i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe

*Graphical illustration: Figure S.4.1
margins, dydx(Lead Warmth) at(ideology=(0 (0.05) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of cand. traits on feeling therm. ratings) title("") scheme(s2color)

*** Predicting vote choice: Reported in Table S.4.1, Model B)
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence##c.ideology c.Moral##c.ideology i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, fe
xtlogit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence##c.ideology c.Moral##c.ideology i.cand_party##c.party7 i.cand_party##ib(2008).VCF0004, or fe

*Graphical illustration: Figure S.4.2
margins, dydx(Lead Warmth) at(ideology=(0 (0.05) 1.0))  level(95) noestimcheck 
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Political Ideology) ytitle(Marginal effect of cand. traits on vote choice) title("") scheme(s2color)


******ROBUSTNESS TEST 4: Difference score approach
// Reported in ONLINE SUPPLEMENTARY MATERIALS S.5

// See separate do-file "Study 1 (ANES) - S5, difference score approach"



******ROBUSTNESS TEST 5: YEAR AND CANDIDATE SPECIFIC MODELS
// Reported in ONLINE SUPPLEMENTARY MATERIALS S.6 (Tables S.6.1 - S.6.7 & Figure S.6.1)

*------------------------------------------------------ 1984 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** Reagan
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1984 & cand_party == 0

*** Mondale
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1984 & cand_party == 1

**** Predicting vote choice
*** Reagan
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1984 & cand_party == 0

*** Mondale
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1984 & cand_party == 1


*------------------------------------------------------ 1988 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** Bush Sr.
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1988 & cand_party == 0

*** Dukakis
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1988 & cand_party == 1

**** Predicting vote choice
*** Bush Sr.
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1988 & cand_party == 0

*** Dukakis
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1988 & cand_party == 1


*------------------------------------------------------ 1992 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** Bush Sr.
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1992 & cand_party == 0

*** Clinton
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1992 & cand_party == 1

**** Predicting vote choice
*** Bush Sr.
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1992 & cand_party == 0

*** Clinton
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1992 & cand_party == 1


*------------------------------------------------------ 1996 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** Dole
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1996 & cand_party == 0

*** Clinton
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1996 & cand_party == 1

**** Predicting vote choice
*** Dole
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1996 & cand_party == 0

*** Clinton
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 1996 & cand_party == 1


*------------------------------------------------------ 2000 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** Bush Jr
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2000 & cand_party == 0

*** Gore
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2000 & cand_party == 1

**** Predicting vote choice
*** Bush Jr
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2000 & cand_party == 0

*** Gore
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2000 & cand_party == 1


*------------------------------------------------------ 2004 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** Bush Jr
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2004 & cand_party == 0

*** Kerry
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2004 & cand_party == 1

**** Predicting vote choice
*** Bush Jr
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2004 & cand_party == 0

*** Kerry
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2004 & cand_party == 1


*------------------------------------------------------ 2008 -----------------------------------------------------------------------------*
**** Predicting feeling thermometer ratings
*** McCain
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2008 & cand_party == 0

*** Obama
reg Feel c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2008 & cand_party == 1

**** Predicting vote choice
*** McCain
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2008 & cand_party == 0

*** Obama
logit vote c.Lead##c.ideology c.Warmth##c.ideology c.Competence c.Moral i.gender c.age c.party7 c.income i.race c.church if VCF0004 == 2008 & cand_party == 1
